
Code continues on the other side ≫
// PROJECT  :SerialEchoAssembly.S
[bookmark: _GoBack]// PURPOSE  :AVR-GCC Assembly USART usage: echo characters to Serial Mon.
// RESULT   :Entered characters echoed on the Serial Monitor  
// COURSE   :ICS4U-E
// AUTHOR   :C. D'Arcy
// DATE     :Created: 2025 02 09.  Confirmed: 2025 03 21.
// MCU      :328P
// STATUS   :Working.
// REFERENCE:Golam Mostafa (halfway down...)
// https://forum.arduino.cc/t/problems-in-serial-communication-with-registers/1016408/6
// P.10. https://raw.githubusercontent.com/abcminiuser/avr-tutorials/master/USART/Output/USART.pdf
// https://hekilledmywire.wordpress.com/2011/01/05/using-the-usartserial-tutorial-part-2/
// http://darcy.rsgc.on.ca/ACES/Datasheets/ATMEGA328P.pdf (Chapter 20)

.global main

#include "avr/io.h"

//The next two lines are the preferred strategy but I kept running into error: Missing ')' Bracket
//#define USART_BAUDRATE 9600        //Desired baud rate
//#define BAUD_PRESCALER (((F_CPU / (USART_BAUDRATE * 16UL))) - 1)
//Hard code the BAUD value to be loaded into the UBRRn register pair
.equ BAUD_PRESCALER, 103         ;precomputed from formula above
.equ data, 65                    ;ASCII value 65 translates to 'A'
.equ ASYNCHRONOUS, 0<<UMSEL01 | 0<<UMSEL00
.equ SYNCHRONOUS,  0<<UMSEL01 | 1<<UMSEL00
.equ PARITY_NONE,  0<<UPM01   | 0<<UPM00
.equ PARITY_EVEN,  1<<UPM01   | 0<<UPM00
.equ PARITY_ODD,   1<<UPM01   | 1<<UPM00
.equ CHAR_SIZE8,   0<<UCSZ02  | 1<<UCSZ01 | 1<<UCSZ00   ;8-bit
.equ STOP_BITS1,   0<<USBS0
.equ STOP_BITS2,   1<<USBS0
.org  0x0000       ;base address of Interrupt Vector (jump) Table (IVT)
reset:
  rjmp   main            

.org  0x0100       ;locate the start of the code well clear of the IVT 
main:
  rcall   init_USART      ;initialize baud, char size, parity, stop bits 
  ldi     r24,data        ;Sample: an ASCII character to be transmitted
  rcall   TXCHAR          ;echo (transmit) it
  rcall   RXCHAR          ;Monitor the input (receive) data.... 
  rcall   TXCHAR          ;echo (transmit) it
  rjmp    .-8             ;infinite loop: RXCHAR > TXCHAR > RXCHAR >...  ret




init_USART:
  ldi   r16,hi8(BAUD_PRESCALER)       ;set the baud rate
  sts   UBRR0H,r16                    ;"
  ldi   r16,lo8(BAUD_PRESCALER)       ;"
  sts   UBRR0L,r16                    ;"

  ldi   r16,(1<<RXEN0) | (1<<TXEN0)   ;enable receive and transmit
  sts   UCSR0B,r16                    ;"
  
  ldi   r16, ASYNCHRONOUS | CHAR_SIZE8 | PARITY_NONE | STOP_BITS1
  sts   UCSR0C,r16                    ;set Comm. Mode  Default: 8N1 
  ret

// Precondition:  Character data (1 byte) to be transmitted is in R24
// Postcondition: Contents of R24 are transmitted via the USART (should appear on the Serial Monitor)
TXCHAR:                               
  lds   r16,UCSR0A          ;determine if the transmit buffer is empty
  sbrs  r16,UDRE0           ;if the flag is set, exit the loop
  rjmp  .-8                 ;loop back to the previous statement 
  sts   UDR0, r24           ;transmit data in r24
  ret

// Precondition:  None
// Postcondition: Character received in the input buffer is stored in R24
RXCHAR:
  lds   r16,UCSR0A          ;
  sbrs  r16,RXC0            ;
  rjmp  .-8                 ;
  lds   r24,UDR0            ;
  ret	

